3x3 MAGIC SQUARES: CUBICAL AND AFFINE

Constituents of the counting functions

and initial values of the functions,

by geometry and generating functions

The number of terms to evaluate in the series expansions, to find the actual numbers of squares and symmetry types, M_c(t), m_c(t), M_a(t), m_a(t).

```
> enddegree:=100;
```

$$enddegree := 100$$

Magic squares: cubical count

The dimension and period for cubical magic, and the standard denominator:

```
> d:=3;
  p:=12;
  cdenom:=(1-x^p)^(d+1);
```

$$d := 3$$

$$p := 12$$

$$cdenom := (1 - x^{12})^4$$

See Section 2.1 of "Six Little Squares". We get LattE to spit out two generating functions--one for the polytope P_c:

```
> ehrpc:=((1) / ((((1) + (-1)*t^1) * (((1) + (-1)*t^1) * (((1) + (-1)*t^2)
  * ((1) + (-1)*t^4))))^1));
```

$$ehrpc := \frac{1}{(1 - t)^2 (1 - t^2) (1 - t^4)}$$

... and one for h intersect P_c:

```
> ehrhc:=((1) / ((((1) + (-1)*t^1) * (((1) + (-1)*t^1) * ((1) +
  (-1)*t^6)))^1));
```

$$ehrhc := \frac{1}{(1 - t)^2 (1 - t^6)}$$

LattE gives the closed generating functions, which we need to convert into their open counterparts:

```
> openehrpc:=simplify(subs(t=1/x,ehrpc));
```

$$openehrpc := \frac{x^8}{(x-1)^2 (x^2-1)(x^4-1)}$$

```
> openehrhc:=simplify(-subs(t=1/x,ehrhc));
```

$$openehrhc := -\frac{x^8}{(x-1)^2 (x^6-1)}$$

By IOP theory, the generating function for the magic quasipolynomial is

```
> mcgfsum:=openehrpc-openehrhc;
  Mcgfsum:=8*mcgfsum;
```

$$mcgfsum := \frac{x^8}{(x-1)^2 (x^2-1)(x^4-1)} + \frac{x^8}{(x-1)^2 (x^6-1)}$$

$$Mcgfsum := \frac{8\,x^8}{(x-1)^2 (x^2-1)(x^4-1)} + \frac{8\,x^8}{(x-1)^2 (x^6-1)}$$

```
> Mcgfs:=simplify(Mcgfsum);
  mcgfs:=simplify(mcgfsum);
```

$$Mcgfs := \frac{8\,x^{10} (2\,x^2+1)}{(x^6-1)(x^4-1)(x-1)^2}$$

$$mcgfs := \frac{x^{10} (2\,x^2+1)}{(x^6-1)(x^4-1)(x-1)^2}$$

Now we bring the total generating function into "normalized form" with denominator "cdenom"...

```
> num:=numer(Mcgfs):
  den:=denom(Mcgfs):
> Magicnum:=expand(simplify(cdenom/den)*num):
  Mcgf:=Magicnum/cdenom;
```

$$Mcgf := \frac{1}{(1-x^{12})^4} (320\,x^{18} + 40\,x^{12} + 128\,x^{15} + 80\,x^{45} + 16\,x^{48} + 896\,x^{24} + 608\,x^{21} + 776\,x^{36} + 1184\,x^{30} + 1040\,x^{33} + 1136\,x^{27} + 464\,x^{39} + 224\,x^{42} + 8\,x^{10} + 16\,x^{11} + 712\,x^{22} + 504\,x^{20} + 400\,x^{19} + 240\,x^{17} + 184\,x^{16} + 96\,x^{14} + 64\,x^{13} + 1168\,x^{28} + 976\,x^{25} + 1056\,x^{26} + 816\,x^{23} + 952\,x^{34} + 1104\,x^{32} + 1168\,x^{31} + 1200\,x^{29} + 864\,x^{35} + 576\,x^{38} + 688\,x^{37} + 56\,x^{46} + 376\,x^{40} + 160\,x^{43} + 120\,x^{44} + 288\,x^{41} + 32\,x^{47})$$

And the symmetry-type generating function ...

```
> num:=numer(mcgfs):
  den:=denom(mcgfs):
> magicnum:=(expand(simplify(cdenom/den)*num)):
  mcgf:=magicnum/cdenom;
```

$$mcgf := \frac{1}{(1-x^{12})^4} (40\,x^{18} + 5\,x^{12} + 16\,x^{15} + 10\,x^{45} + 2\,x^{48} + 112\,x^{24} + 76\,x^{21} + 97\,x^{36} + 148\,x^{30} + 130\,x^{33} + 142\,x^{27} + 58\,x^{39} + 28\,x^{42} + x^{10} + 2\,x^{11} + 89\,x^{22} + 63\,x^{20} + 50\,x^{19} + 30\,x^{17} + 23\,x^{16}$$

$+ 12 x^{14} + 8 x^{13} + 146 x^{28} + 122 x^{25} + 132 x^{26} + 102 x^{23} + 119 x^{34} + 138 x^{32} + 146 x^{31} + 150 x^{29} + 108 x^{35} + 72 x^{38} + 86 x^{37} + 7 x^{46} + 47 x^{40} + 20 x^{43} + 15 x^{44} + 36 x^{41} + 4 x^{47})$

Let's check the first few terms...

```
> Mcgfseries:=series(Mcgf,x=0,enddegree+1);
```

$$Mcgfseries := 8 x^{10} + 16 x^{11} + 40 x^{12} + 64 x^{13} + 96 x^{14} + 128 x^{15} + 184 x^{16} + 240 x^{17} + 320 x^{18} + 400 x^{19} + 504 x^{20} + 608 x^{21} + 744 x^{22} + 880 x^{23} + 1056 x^{24} + 1232 x^{25} + 1440 x^{26} + 1648 x^{27} + 1904 x^{28} + 2160 x^{29} + 2464 x^{30} + 2768 x^{31} + 3120 x^{32} + 3472 x^{33} + 3880 x^{34} + 4288 x^{35} + 4760 x^{36} + 5232 x^{37} + 5760 x^{38} + 6288 x^{39} + 6888 x^{40} + 7488 x^{41} + 8160 x^{42} + 8832 x^{43} + 9576 x^{44} + 10320 x^{45} + 11144 x^{46} + 11968 x^{47} + 12880 x^{48} + 13792 x^{49} + 14784 x^{50} + 15776 x^{51} + 16864 x^{52} + 17952 x^{53} + 19136 x^{54} + 20320 x^{55} + 21600 x^{56} + 22880 x^{57} + 24264 x^{58} + 25648 x^{59} + 27144 x^{60} + 28640 x^{61} + 30240 x^{62} + 31840 x^{63} + 33560 x^{64} + 35280 x^{65} + 37120 x^{66} + 38960 x^{67} + 40920 x^{68} + 42880 x^{69} + 44968 x^{70} + 47056 x^{71} + 49280 x^{72} + 51504 x^{73} + 53856 x^{74} + 56208 x^{75} + 58704 x^{76} + 61200 x^{77} + 63840 x^{78} + 66480 x^{79} + 69264 x^{80} + 72048 x^{81} + 74984 x^{82} + 77920 x^{83} + 81016 x^{84} + 84112 x^{85} + 87360 x^{86} + 90608 x^{87} + 94024 x^{88} + 97440 x^{89} + 101024 x^{90} + 104608 x^{91} + 108360 x^{92} + 112112 x^{93} + 116040 x^{94} + 119968 x^{95} + 124080 x^{96} + 128192 x^{97} + 132480 x^{98} + 136768 x^{99} + 141248 x^{100} + O(x^{101})$$

```
> mcgfseries:=series(mcgf,x=0,enddegree+1);
```

$$mcgfseries := x^{10} + 2 x^{11} + 5 x^{12} + 8 x^{13} + 12 x^{14} + 16 x^{15} + 23 x^{16} + 30 x^{17} + 40 x^{18} + 50 x^{19} + 63 x^{20} + 76 x^{21} + 93 x^{22} + 110 x^{23} + 132 x^{24} + 154 x^{25} + 180 x^{26} + 206 x^{27} + 238 x^{28} + 270 x^{29} + 308 x^{30} + 346 x^{31} + 390 x^{32} + 434 x^{33} + 485 x^{34} + 536 x^{35} + 595 x^{36} + 654 x^{37} + 720 x^{38} + 786 x^{39} + 861 x^{40} + 936 x^{41} + 1020 x^{42} + 1104 x^{43} + 1197 x^{44} + 1290 x^{45} + 1393 x^{46} + 1496 x^{47} + 1610 x^{48} + 1724 x^{49} + 1848 x^{50} + 1972 x^{51} + 2108 x^{52} + 2244 x^{53} + 2392 x^{54} + 2540 x^{55} + 2700 x^{56} + 2860 x^{57} + 3033 x^{58} + 3206 x^{59} + 3393 x^{60} + 3580 x^{61} + 3780 x^{62} + 3980 x^{63} + 4195 x^{64} + 4410 x^{65} + 4640 x^{66} + 4870 x^{67} + 5115 x^{68} + 5360 x^{69} + 5621 x^{70} + 5882 x^{71} + 6160 x^{72} + 6438 x^{73} + 6732 x^{74} + 7026 x^{75} + 7338 x^{76} + 7650 x^{77} + 7980 x^{78} + 8310 x^{79} + 8658 x^{80} + 9006 x^{81} + 9373 x^{82} + 9740 x^{83} + 10127 x^{84} + 10514 x^{85} + 10920 x^{86} + 11326 x^{87} + 11753 x^{88} + 12180 x^{89} + 12628 x^{90} + 13076 x^{91} + 13545 x^{92} + 14014 x^{93} + 14505 x^{94} + 14996 x^{95} + 15510 x^{96} + 16024 x^{97} + 16560 x^{98} + 17096 x^{99} + 17656 x^{100} + O(x^{101})$$

So here are the constituents of the cubic total-count quasipolynomial (starting with the 0th constituent); also with denominator cleared and factored when possible:

```
> Mcconstituent[0]:=expand(sum(coeff(Magicnum,x,p*j)*binomial(d+t/p-j,d),j
  =1..d+1));
  6*%;
  factor(%);
```

$$Mcconstituent_0 := \frac{38}{3} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - 16$$

$$76 t - 16 t^2 + t^3 - 96$$

$$(t - 6) (t - 2) (t - 8)$$

```
> for r from 1 to p do

  Mcconstituent[r]:=expand(sum(coeff(Magicnum,x,p*j+r)*binomial(d+(t-r)/p-
```

```
j,d),j=0..d));
  6*Mcconstituent[r];
  factor(%);
od;
```

$$Mcconstituent_1 := \frac{73}{6} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - \frac{29}{3}$$

$$73\, t - 16\, t^2 + t^3 - 58$$

$$(t - 1)\, (t^2 - 15\, t + 58)$$

$$Mcconstituent_2 := \frac{38}{3} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - 16$$

$$76\, t - 16\, t^2 + t^3 - 96$$

$$(t - 6)\, (t - 2)\, (t - 8)$$

$$Mcconstituent_3 := \frac{73}{6} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - 17$$

$$73\, t - 16\, t^2 + t^3 - 102$$

$$(t - 3)\, (t^2 - 13\, t + 34)$$

$$Mcconstituent_4 := \frac{38}{3} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - \frac{56}{3}$$

$$76\, t - 16\, t^2 + t^3 - 112$$

$$(t - 4)\, (t^2 - 12\, t + 28)$$

$$Mcconstituent_5 := \frac{73}{6} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - 15$$

$$73\, t - 16\, t^2 + t^3 - 90$$

$$(t - 5)\, (t - 2)\, (t - 9)$$

$$Mcconstituent_6 := \frac{38}{3} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - 16$$

$$76\, t - 16\, t^2 + t^3 - 96$$

$$(t - 6)\, (t - 2)\, (t - 8)$$

$$Mcconstituent_7 := \frac{73}{6} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - \frac{35}{3}$$

$$73\, t - 16\, t^2 + t^3 - 70$$

$$(t - 7)\, (t^2 - 9\, t + 10)$$

$$Mcconstituent_8 := \frac{38}{3} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - 16$$

$$76\, t - 16\, t^2 + t^3 - 96$$

$$(t - 6)\, (t - 2)\, (t - 8)$$

$$Mcconstituent_9 := \frac{73}{6} t - \frac{8}{3} t^2 + \frac{1}{6} t^3 - 15$$

$$73\,t - 16\,t^2 + t^3 - 90$$
$$(t-5)\,(t-2)\,(t-9)$$
$$Mconstituent_{10} := \frac{38}{3}\,t - \frac{8}{3}\,t^2 + \frac{1}{6}\,t^3 - \frac{56}{3}$$
$$76\,t - 16\,t^2 + t^3 - 112$$
$$(t-4)\,(t^2 - 12\,t + 28)$$
$$Mconstituent_{11} := \frac{73}{6}\,t - \frac{8}{3}\,t^2 + \frac{1}{6}\,t^3 - 17$$
$$73\,t - 16\,t^2 + t^3 - 102$$
$$(t-3)\,(t^2 - 13\,t + 34)$$
$$Mconstituent_{12} := \frac{38}{3}\,t - \frac{8}{3}\,t^2 + \frac{1}{6}\,t^3 - 16$$
$$76\,t - 16\,t^2 + t^3 - 96$$
$$(t-6)\,(t-2)\,(t-8)$$

And here are the constituents of the cubic symmetry-class quasipolynomial (starting with the 0th constituent) (which will be 1/8 those of the total count):

```
> mcconstituent[0]:=expand(sum(coeff(magicnum,x,p*j)*binomial(d+t/p-j,d),j
  =1..d+1));
```

$$mcconstituent_0 := \frac{19}{12}\,t - \frac{1}{3}\,t^2 + \frac{1}{48}\,t^3 - 2$$

```
> for r from 1 to p do

  mcconstituent[r]:=expand(sum(coeff(magicnum,x,p*j+r)*binomial(d+(t-r)/p-
  j,d),j=0..d));
  od;
```

$$mcconstituent_1 := \frac{73}{48}\,t - \frac{1}{3}\,t^2 + \frac{1}{48}\,t^3 - \frac{29}{24}$$
$$mcconstituent_2 := \frac{19}{12}\,t - \frac{1}{3}\,t^2 + \frac{1}{48}\,t^3 - 2$$
$$mcconstituent_3 := \frac{73}{48}\,t - \frac{1}{3}\,t^2 + \frac{1}{48}\,t^3 - \frac{17}{8}$$
$$mcconstituent_4 := \frac{19}{12}\,t - \frac{1}{3}\,t^2 + \frac{1}{48}\,t^3 - \frac{7}{3}$$
$$mcconstituent_5 := \frac{73}{48}\,t - \frac{1}{3}\,t^2 + \frac{1}{48}\,t^3 - \frac{15}{8}$$
$$mcconstituent_6 := \frac{19}{12}\,t - \frac{1}{3}\,t^2 + \frac{1}{48}\,t^3 - 2$$
$$mcconstituent_7 := \frac{73}{48}\,t - \frac{1}{3}\,t^2 + \frac{1}{48}\,t^3 - \frac{35}{24}$$

$$mcconstituent_8 := \frac{19}{12}t - \frac{1}{3}t^2 + \frac{1}{48}t^3 - 2$$

$$mcconstituent_9 := \frac{73}{48}t - \frac{1}{3}t^2 + \frac{1}{48}t^3 - \frac{15}{8}$$

$$mcconstituent_{10} := \frac{19}{12}t - \frac{1}{3}t^2 + \frac{1}{48}t^3 - \frac{7}{3}$$

$$mcconstituent_{11} := \frac{73}{48}t - \frac{1}{3}t^2 + \frac{1}{48}t^3 - \frac{17}{8}$$

$$mcconstituent_{12} := \frac{19}{12}t - \frac{1}{3}t^2 + \frac{1}{48}t^3 - 2$$

Test the cubic total-count constituents up to "enddegree". Print any n that give wrong evaluation.

```
> print("Test cubic constituents; print any inconsistent evaluations.");
  for n from 1 to enddegree do
    r:=modp(n,p);
    value:=eval(Mcconstituent[r],t=n);
    term:=coeff(Mcgfseries,x,n);
    if ( value-term <> 0 ) then print(n,r,value,term,value-term); fi;
  od:
  print("Values checked from 1 to", enddegree);
```

"Test cubic constituents; print any inconsistent evaluations."

"Values checked from 1 to", 100

Reduced magic squares: cubic count

First, the generating functions of R_mc(n) and r_mc(n). The conversion factor is:

```
> rmultiplier:=x^2/(1-x)^2:
```

```
> Rgfs:=simplify(Mcgfs/rmultiplier);
  rgfs:=simplify(mcgfs/rmultiplier);
```

$$Rgfs := \frac{8\,(2\,x^2+1)\,x^8}{(x-1)^2\,(x^3+x^2+x+1)\,(x^5+x^4+x^3+x^2+x+1)}$$

$$rgfs := \frac{(2\,x^2+1)\,x^8}{(x-1)^2\,(x^3+x^2+x+1)\,(x^5+x^4+x^3+x^2+x+1)}$$

Now standardize the denominator. dr = the degree for reduced squares.

```
> dr:=1;
  rdenom:=(1-x^p)^(dr+1);
> num:=numer(Rgfs):
  den:=denom(Rgfs):
> Rmagicnum:=simplify(rdenom/den)*num:
  Rgf:=Rmagicnum/rdenom;
> num:=numer(rgfs):
  den:=denom(rgfs):
> rmagicnum:=simplify(rdenom/den)*num:
```

```
rgf:=rmagicnum/rdenom;
```

$$dr := 1$$

$$rdenom := (1 - x^{12})^2$$

$$Rgf := \frac{8\,(x^{10} + x^8 + x^6 + x^4 + x^2 + 1)\,(x^4 - x^2 + 1)\,(2\,x^2 + 1)\,x^8}{(1 - x^{12})^2}$$

$$rgf := \frac{(x^{10} + x^8 + x^6 + x^4 + x^2 + 1)\,(x^4 - x^2 + 1)\,(2\,x^2 + 1)\,x^8}{(1 - x^{12})^2}$$

Let's check the first few terms...

```
> Rgfseries:=series(Rgf,x=0,enddegree+1);
```

$$Rgfseries := 8\,x^8 + 16\,x^{10} + 8\,x^{12} + 24\,x^{14} + 24\,x^{16} + 24\,x^{18} + 32\,x^{20} + 40\,x^{22} + 32\,x^{24} + 48\,x^{26} + 48\,x^{28} + 48\,x^{30} + 56\,x^{32} + 64\,x^{34} + 56\,x^{36} + 72\,x^{38} + 72\,x^{40} + 72\,x^{42} + 80\,x^{44} + 88\,x^{46} + 80\,x^{48} + 96\,x^{50} + 96\,x^{52} + 96\,x^{54} + 104\,x^{56} + 112\,x^{58} + 104\,x^{60} + 120\,x^{62} + 120\,x^{64} + 120\,x^{66} + 128\,x^{68} + 136\,x^{70} + 128\,x^{72} + 144\,x^{74} + 144\,x^{76} + 144\,x^{78} + 152\,x^{80} + 160\,x^{82} + 152\,x^{84} + 168\,x^{86} + 168\,x^{88} + 168\,x^{90} + 176\,x^{92} + 184\,x^{94} + 176\,x^{96} + 192\,x^{98} + 192\,x^{100} + O(x^{102})$$

```
> rgfseries:=series(rgf,x=0,enddegree+1);
```

$$rgfseries := x^8 + 2\,x^{10} + x^{12} + 3\,x^{14} + 3\,x^{16} + 3\,x^{18} + 4\,x^{20} + 5\,x^{22} + 4\,x^{24} + 6\,x^{26} + 6\,x^{28} + 6\,x^{30} + 7\,x^{32} + 8\,x^{34} + 7\,x^{36} + 9\,x^{38} + 9\,x^{40} + 9\,x^{42} + 10\,x^{44} + 11\,x^{46} + 10\,x^{48} + 12\,x^{50} + 12\,x^{52} + 12\,x^{54} + 13\,x^{56} + 14\,x^{58} + 13\,x^{60} + 15\,x^{62} + 15\,x^{64} + 15\,x^{66} + 16\,x^{68} + 17\,x^{70} + 16\,x^{72} + 18\,x^{74} + 18\,x^{76} + 18\,x^{78} + 19\,x^{80} + 20\,x^{82} + 19\,x^{84} + 21\,x^{86} + 21\,x^{88} + 21\,x^{90} + 22\,x^{92} + 23\,x^{94} + 22\,x^{96} + 24\,x^{98} + 24\,x^{100} + O(x^{102})$$

Magic squares: affine count

The dimension and period for affine magic, and the standard denominator:

```
> da:=2;
  pa:=18;
  adenom:=(1-x^pa)^(da+1);
```

$$da := 2$$

$$pa := 18$$

$$adenom := (1 - x^{18})^3$$

See Section 2.2 of "Six Little Squares". We get LattE to spit out two generating functions--one for the polytope P_a:

```
> magicaffp:=(1-t)/(1-t^3) * ((1) / ((((1) + (-1)*t^1) * (((1) + (-1)*t^3)
  * ((1) + (-1)*t^6)))^1));
```

$$magicaffp := \frac{1}{(1 - t^3)^2\,(1 - t^6)}$$

... and one for P_a intersect h:

```
> magicaffh:=(1-t)/(1-t^3) * ((1) / ((((1) + (-1)*t^1) * ((1) +
  (-1)*t^9))^1));
```

$$magicaffh := \frac{1}{(1-t^3)(1-t^9)}$$

From this point on we proceed as above...

```
> openmagicaffp:=simplify(-subs(t=1/x,magicaffp));
```

$$openmagicaffp := -\frac{x^{12}}{(x^3-1)^2(x^6-1)}$$

```
> openmagicaffh:=simplify(subs(t=1/x,magicaffh));
```

$$openmagicaffh := \frac{x^{12}}{(x^3-1)(x^9-1)}$$

```
> magfsum:=openmagicaffp-openmagicaffh;
  Magfsum:=8*magfsum;
```

$$magfsum := -\frac{x^{12}}{(x^3-1)^2(x^6-1)} - \frac{x^{12}}{(x^3-1)(x^9-1)}$$

$$Magfsum := -\frac{8\,x^{12}}{(x^3-1)^2(x^6-1)} - \frac{8\,x^{12}}{(x^3-1)(x^9-1)}$$

```
> Magfs:=simplify(Magfsum);
  magfs:=simplify(magfsum);
```

$$Magfs := -\frac{8\,(2\,x^3+1)\,x^{15}}{(x^9-1)(x^3-1)(x^6-1)}$$

$$magfs := -\frac{(2\,x^3+1)\,x^{15}}{(x^9-1)(x^3-1)(x^6-1)}$$

Now we bring the total generating function into "normalized form" with denominator "cdenom"...

```
> num:=numer(Magfs):
  den:=denom(Magfs):
> Magicnumaff:=expand(simplify(adenom/den)*num):
  Magf:=Magicnumaff/adenom;
```

$$Magf := \frac{1}{(1-x^{18})^3}(8\,x^{15} + 112\,x^{33} + 80\,x^{27} + 24\,x^{18} + 32\,x^{21} + 56\,x^{24} + 104\,x^{30} + 112\,x^{39} + 104\,x^{36} + 24\,x^{51} + 64\,x^{45} + 40\,x^{48} + 88\,x^{42} + 16\,x^{54})$$

And the symmetry-type generating function ...

```
> num:=numer(magfs):
  den:=denom(magfs):
> magicnumaff:=(expand(simplify(adenom/den)*num)):
  magf:=magicnumaff/adenom;
```

$$magf := \frac{1}{(1-x^{18})^3}(x^{15} + 14x^{33} + 10x^{27} + 3x^{18} + 4x^{21} + 7x^{24} + 13x^{30} + 14x^{39} + 13x^{36} + 3x^{51} + 8x^{45} + 5x^{48} + 11x^{42} + 2x^{54})$$

Now we expand the series to find the first few values of the counts.

```
> Magfseries:=series(Magf,x=0,enddegree+1);
```

$$Magfseries := 8x^{15} + 24x^{18} + 32x^{21} + 56x^{24} + 80x^{27} + 104x^{30} + 136x^{33} + 176x^{36} + 208x^{39} + 256x^{42} + 304x^{45} + 352x^{48} + 408x^{51} + 472x^{54} + 528x^{57} + 600x^{60} + 672x^{63} + 744x^{66} + 824x^{69} + 912x^{72} + 992x^{75} + 1088x^{78} + 1184x^{81} + 1280x^{84} + 1384x^{87} + 1496x^{90} + 1600x^{93} + 1720x^{96} + 1840x^{99} + O(x^{102})$$

```
> magfseries:=series(magf,x=0,enddegree+1);
```

$$magfseries := x^{15} + 3x^{18} + 4x^{21} + 7x^{24} + 10x^{27} + 13x^{30} + 17x^{33} + 22x^{36} + 26x^{39} + 32x^{42} + 38x^{45} + 44x^{48} + 51x^{51} + 59x^{54} + 66x^{57} + 75x^{60} + 84x^{63} + 93x^{66} + 103x^{69} + 114x^{72} + 124x^{75} + 136x^{78} + 148x^{81} + 160x^{84} + 173x^{87} + 187x^{90} + 200x^{93} + 215x^{96} + 230x^{99} + O(x^{102})$$

The generating functions of R_mc(n) and r_mc(n). The conversion factor is:

```
> ramultiplier:=x^3/(1-x^3);
```

$$ramultiplier := \frac{x^3}{1-x^3}$$

```
> Ragfs:=simplify(Magfs/ramultiplier);
  ragfs:=simplify(magfs/ramultiplier);
```

$$Ragfs := \frac{8x^{12}(2x^3+1)}{(x^6-1)(x^3-1)(x^6+x^3+1)}$$

$$ragfs := \frac{x^{12}(2x^3+1)}{(x^6-1)(x^3-1)(x^6+x^3+1)}$$

Now standardize the denominator for reduced g.f. dra = the degree for reduced squares.

```
> dra:=1;
  radenom:=(1-x^pa)^(dra+1);
> num:=numer(Ragfs):
  den:=denom(Ragfs):
> Ramagicnum:=simplify(radenom/den)*num:
  Ragf:=Ramagicnum/radenom;
> num:=numer(ragfs):
  den:=denom(ragfs):
> ramagicnum:=simplify(radenom/den)*num:
  ragf:=ramagicnum/radenom;
```

$$dra := 1$$

$$radenom := (1-x^{18})^2$$

$$Ragf := \frac{8(x^{12}+x^6+1)(x^9+1)x^{12}(2x^3+1)}{(1-x^{18})^2}$$

$$ragf := \frac{(x^{12} + x^6 + 1)(x^9 + 1) x^{12} (2x^3 + 1)}{(1 - x^{18})^2}$$

Let's check the first few terms of the reduced series...

```
> Ragfseries:=series(Ragf,x=0,enddegree+1);
```

$$Ragfseries := 8x^{12} + 16x^{15} + 8x^{18} + 24x^{21} + 24x^{24} + 24x^{27} + 32x^{30} + 40x^{33} + 32x^{36} + 48x^{39} + 48x^{42} + 48x^{45} + 56x^{48} + 64x^{51} + 56x^{54} + 72x^{57} + 72x^{60} + 72x^{63} + 80x^{66} + 88x^{69} + 80x^{72} + 96x^{75} + 96x^{78} + 96x^{81} + 104x^{84} + 112x^{87} + 104x^{90} + 120x^{93} + 120x^{96} + 120x^{99} + O(x^{102})$$

```
> ragfseries:=series(ragf,x=0,enddegree+1);
```

$$ragfseries := x^{12} + 2x^{15} + x^{18} + 3x^{21} + 3x^{24} + 3x^{27} + 4x^{30} + 5x^{33} + 4x^{36} + 6x^{39} + 6x^{42} + 6x^{45} + 7x^{48} + 8x^{51} + 7x^{54} + 9x^{57} + 9x^{60} + 9x^{63} + 10x^{66} + 11x^{69} + 10x^{72} + 12x^{75} + 12x^{78} + 12x^{81} + 13x^{84} + 14x^{87} + 13x^{90} + 15x^{93} + 15x^{96} + 15x^{99} + O(x^{102})$$

The constituents of the affine total counting function. First, the 0th constituent, then times 9/2 to clear the denominators and a trivial constant factor.

```
> Maconstituent[0]:=expand(sum(coeff(Magicnumaff,x,pa*j)*binomial(da+t/pa-j,da),j=1..da+1));
  9/2*Maconstituent[0];
  factor(%);
```

$$Maconstituent_0 := -\frac{32}{9}t + \frac{2}{9}t^2 + 16$$

$$-16t + t^2 + 72$$

$$-16t + t^2 + 72$$

The constituents, then times 9/2 to clear denominators and a trivial constant factor; also, factored if possible.

```
> for r from 1 to pa do

  Maconstituent[r]:=expand(sum(coeff(Magicnumaff,x,pa*j+r)*binomial(da+(t-r)/pa-j,da),j=0..d));
    9/2*%;
    factor(%);
  od;
```

$$Maconstituent_1 := 0$$

$$0$$

$$0$$

$$Maconstituent_2 := 0$$

$$0$$

$$0$$

$$Maconstituent_3 := -\frac{32}{9}t + \frac{2}{9}t^2 + \frac{26}{3}$$

$$-16t + t^2 + 39$$

$$(t - 3)(t - 13)$$

$$Maconstituent_4 := 0$$

$$0$$
$$0$$
$$Maconstituent_5 := 0$$
$$0$$
$$0$$
$$Maconstituent_6 := -\frac{32}{9}t + \frac{2}{9}t^2 + \frac{40}{3}$$
$$-16\,t + t^2 + 60$$
$$(t - 6)\,(t - 10)$$
$$Maconstituent_7 := 0$$
$$0$$
$$0$$
$$Maconstituent_8 := 0$$
$$0$$
$$0$$
$$Maconstituent_9 := \frac{2}{9}t^2 + 14 - \frac{32}{9}t$$
$$t^2 + 63 - 16\,t$$
$$(t - 7)\,(t - 9)$$
$$Maconstituent_{10} := 0$$
$$0$$
$$0$$
$$Maconstituent_{11} := 0$$
$$0$$
$$0$$
$$Maconstituent_{12} := -\frac{32}{9}t + \frac{2}{9}t^2 + \frac{32}{3}$$
$$-16\,t + t^2 + 48$$
$$(t - 4)\,(t - 12)$$
$$Maconstituent_{13} := 0$$
$$0$$
$$0$$
$$Maconstituent_{14} := 0$$
$$0$$
$$0$$
$$Maconstituent_{15} := \frac{34}{3} - \frac{32}{9}t + \frac{2}{9}t^2$$

$$51 - 16\,t + t^2$$

$$51 - 16\,t + t^2$$

$$Maconstituent_{16} := 0$$

$$0$$

$$0$$

$$Maconstituent_{17} := 0$$

$$0$$

$$0$$

$$Maconstituent_{18} := -\frac{32}{9}\,t + \frac{2}{9}\,t^2 + 16$$

$$-16\,t + t^2 + 72$$

$$-16\,t + t^2 + 72$$

The constituents of the affine symmetry-class counting function. First, the 0th constituent.

```
> maconstituent[0]:=expand(sum(coeff(magicnumaff,x,pa*j)*binomial(da+t/pa-
  j,da),j=1..da+1));
```

$$maconstituent_0 := -\frac{4}{9}\,t + \frac{1}{36}\,t^2 + 2$$

The constituents. They should all be 1/8 of the total constituents.

```
> for r from 1 to pa do

  maconstituent[r]:=expand(sum(coeff(magicnumaff,x,pa*j+r)*binomial(da+(t-
  r)/pa-j,da),j=0..d));
  od;
```

$$maconstituent_1 := 0$$

$$maconstituent_2 := 0$$

$$maconstituent_3 := -\frac{4}{9}\,t + \frac{1}{36}\,t^2 + \frac{13}{12}$$

$$maconstituent_4 := 0$$

$$maconstituent_5 := 0$$

$$maconstituent_6 := -\frac{4}{9}\,t + \frac{1}{36}\,t^2 + \frac{5}{3}$$

$$maconstituent_7 := 0$$

$$maconstituent_8 := 0$$

$$maconstituent_9 := \frac{1}{36}\,t^2 + \frac{7}{4} - \frac{4}{9}\,t$$

$$maconstituent_{10} := 0$$

$$maconstituent_{11} := 0$$

$$maconstituent_{12} := -\frac{4}{9}t + \frac{1}{36}t^2 + \frac{4}{3}$$

$$maconstituent_{13} := 0$$

$$maconstituent_{14} := 0$$

$$maconstituent_{15} := \frac{17}{12} - \frac{4}{9}t + \frac{1}{36}t^2$$

$$maconstituent_{16} := 0$$

$$maconstituent_{17} := 0$$

$$maconstituent_{18} := -\frac{4}{9}t + \frac{1}{36}t^2 + 2$$

Test affine total-count constituents up to "enddegree". Print any n that give wrong evaluation.

```
> print("Test affine constituents; print any inconsistent evaluations.");
  for n from 1 to enddegree do
    r:=modp(n,pa);
    value:=eval(Maconstituent[r],t=n);
    term:=coeff(Magfseries,x,n);
    if ( value-term <> 0 ) then print(n,r,value,term,value-term); fi;
  od:
  print("Values checked from 1 to", enddegree);
```

"Test affine constituents; print any inconsistent evaluations."

"Values checked from 1 to", 100

The constituents of the affine reduced counting function. First, the 0th constituent, then times 3/4 to clear the denominators and a trivial constant factor.

```
> Raconstituent[0]:=expand(sum(coeff(Ramagicnum,x,pa*j)*binomial(dra+t/pa-
  j,dra),j=1..dra+1));
  3/4*Raconstituent[0];
```

$$Raconstituent_0 := \frac{4}{3}t - 16$$

$$t - 12$$

The constituents, then times 3/4 to clear denominators and a trivial constant factor.

```
> for r from 1 to pa do

  Raconstituent[r]:=expand(sum(coeff(Ramagicnum,x,pa*j+r)*binomial(dra+(t-
  r)/pa-j,dra),j=0..dra));
    3/4*%;
  od;
```

$$Raconstituent_1 := 0$$

$$0$$

$$Raconstituent_2 := 0$$

$$0$$

$$Raconstituent_3 := \frac{4}{3}t - 4$$

$$t - 3$$

$$Raconstituent_4 := 0$$

$$0$$

$$Raconstituent_5 := 0$$

$$0$$

$$Raconstituent_6 := \frac{4}{3} t - 8$$

$$t - 6$$

$$Raconstituent_7 := 0$$

$$0$$

$$Raconstituent_8 := 0$$

$$0$$

$$Raconstituent_9 := \frac{4}{3} t - 12$$

$$t - 9$$

$$Raconstituent_{10} := 0$$

$$0$$

$$Raconstituent_{11} := 0$$

$$0$$

$$Raconstituent_{12} := \frac{4}{3} t - 8$$

$$t - 6$$

$$Raconstituent_{13} := 0$$

$$0$$

$$Raconstituent_{14} := 0$$

$$0$$

$$Raconstituent_{15} := \frac{4}{3} t - 4$$

$$t - 3$$

$$Raconstituent_{16} := 0$$

$$0$$

$$Raconstituent_{17} := 0$$

$$0$$

$$Raconstituent_{18} := \frac{4}{3} t - 16$$

$$t - 12$$

The constituents of the affine reduced symmetry-class counting function. First, the 0th constituent.

```
> raconstituent[0]:=expand(sum(coeff(ramagicnum,x,pa*j)*binomial(dra+t/pa-
```

```
  j,dra),j=1..dra+1));
```

$$raconstituent_0 := \frac{1}{6}t - 2$$

The constituents. They should all be 1/8 of the reduced constituents.

```
> for r from 1 to pa do

  raconstituent[r]:=expand(sum(coeff(ramagicnum,x,pa*j+r)*binomial(dra+(t-
  r)/pa-j,dra),j=0..dra));
    6*%;
  od;
```

$$raconstituent_1 := 0$$

$$0$$

$$raconstituent_2 := 0$$

$$0$$

$$raconstituent_3 := \frac{1}{6}t - \frac{1}{2}$$

$$t - 3$$

$$raconstituent_4 := 0$$

$$0$$

$$raconstituent_5 := 0$$

$$0$$

$$raconstituent_6 := \frac{1}{6}t - 1$$

$$t - 6$$

$$raconstituent_7 := 0$$

$$0$$

$$raconstituent_8 := 0$$

$$0$$

$$raconstituent_9 := \frac{1}{6}t - \frac{3}{2}$$

$$t - 9$$

$$raconstituent_{10} := 0$$

$$0$$

$$raconstituent_{11} := 0$$

$$0$$

$$raconstituent_{12} := \frac{1}{6}t - 1$$

$$t - 6$$

$$raconstituent_{13} := 0$$

$$0$$

$$raconstituent_{14} := 0$$

$$0$$

$$raconstituent_{15} := \frac{1}{6} t - \frac{1}{2}$$

$$t - 3$$

$$raconstituent_{16} := 0$$

$$0$$

$$raconstituent_{17} := 0$$

$$0$$

$$raconstituent_{18} := \frac{1}{6} t - 2$$

$$t - 12$$

Test affine reduced constituents up to "enddegree". Print any n that give wrong evaluation.

```
> print("Test affine reduced constituents; print any inconsistent
  evaluations.");
  for n from 1 to enddegree do
    r:=modp(n,pa);
    value:=eval(Raconstituent[r],t=n);
    term:=coeff(Ragfseries,x,n);
    if ( value-term <> 0 ) then print(n,r,value,term,value-term); fi;
  od:
  print("Values checked from 1 to", enddegree);
```

"Test affine reduced constituents; print any inconsistent evaluations."

"Values checked from 1 to", 100

Test affine reduced symmetry-type constituents up to "enddegree". Print any n that give wrong evaluation.

```
> print("Test affine reduced symmetry-type constituents; print any
  inconsistent evaluations.");
  for n from 1 to enddegree do
    r:=modp(n,pa);
    value:=eval(raconstituent[r],t=n);
    term:=coeff(ragfseries,x,n);
    if ( value-term <> 0 ) then print(n,r,value,term,value-term); fi;
  od:
  print("Values checked from 1 to", enddegree);
```

"Test affine reduced symmetry-type constituents; print any inconsistent evaluations."

"Values checked from 1 to", 100